Intelligent trash can

Contents

1. Project description

The project consists in an intelligent trash can with the following characteristics:

- It detects motion in front of it using an infrared sensor. The sensor is screwed in the can's front, but beneath the lid.
- When motion is detected the lid is opened with the help of a stepper motor. The motor is screwed in the can's back. The motor's shaft and the lid are interconnected with a wire in order that when the shaft rotates clockwise the lid opens and when it rotates counterclockwise the lid closes.
- It detects when it is fully filled using an ultrasonic sensor. The sensor is screwed inside the can in order to detect the garbage that accumulates.
- When it is fully filled a red LED is turned on. Otherwise the LED is turned off.

2. Hardware description

I used an Arduino Uno R3 board that has an ATmega328P microcontroller. It is a high performance, low power Atmel®AVR® 8-bit microcontroller. It has an advanced RISC architecture which consists of 131 powerful instructions – most are executed in a single clock cycle, 32 general purpose working registers, an on-chip 2-cycle multiplier etc. It also has 32 KB of flash memory, 1 KB of EEPROM, 2 KB of SRAM, a 10-bit ADC, a SPI port etc.



In this project I used the following hardware pieces:

- An GP2Y0A21YK0F IR sensor. It detects motion in front of the can on a distance between 10 and 80cm. The supply is connected to the breadboard while the output is also analog and it is connected to an analog input pin of the board.
- A 5V 28BYJ-48 stepper motor with an ULN2003 driver. It opens the can's lid when the IR sensor returns an appropiate value. The motor is connected to the driver while the driver's supply is connected to the breadboard and the digital inputs are connected to 4 digital I/O pins of the board. The digital I/O pins deliver 5V for logic 1 and 0V for logic 0.
- A HC-SR04 ultrasonic sensor. It detects when the can is fully filled. The supply is connected to the breadboard while the echo input and trigger output are connected to the board.
- A 5mm red LED. It turns on when the ultrasonic sensor returns an appropiate value. It is placed on the breadboard and it is in series with a resistor. The resistor's resistance is 240Ω in

order to pass 12mA through the LED and the voltage on the LED to be 2.12V while the supply for the circuit is delivered by a digital I/O pin of the board.

- A 400 points breadboard. It is supplied by the board with 5V. It supplies the sensors and the driver. It is also the link between the LED and the digital I/O pin of the board that supplies the LED and the afferent resistor.
- 6 male-male and 20 female-male DuPont wires each with a length of 30cm. They are used to make the necessary connections between components.

3. Software description



- Outside the functions I created the object ”stepperMotor” from the class ”Stepper” and this represents the motor. In order to do this I used the default library ”Stepper”. I also created a bool variable titled ”opened” that initially is false in order to represent that when the board is powered up, the can is most likely closed. This variable is used in the ”irSensorAndStepper” function.
- In the ”setup” function I set the baudrate to be 9600 b/s, this value being enough for most Arduino projects, including mine. I need the serial communication to print in the ”Serial Monitor” the output of the IR sensor and the measured distance of the ultrasonic sensor in order to check if the sensors work properly. The pin to which the LED’s anode is connected is set as output in order for the LED to turn on when the can is full. The pins to which the trigger and echo functions of the ultrasonic sensor are connected are set in accordance with the functionality of the sensor.
- In the ”ultrasonicSensorMeasurement” function the ultrasonic sensor sends a pulse and measures the time in which the bounced pulse received on the echo pin is high. There are delays between sending and measuring to not stress the sensor. The measured distance is calculated as the product between the aforementioned time and the speed with which the pulse travels divided by 2 (the pulse is an ultrasonic wave that travels with the speed of sound which is 340.3m/s, but it travels the same distance, therefore the speed is 170.15m/s or 0.017015cm/µs). The distance is printed in the ”Serial Monitor”. If the distance is lesser than

7.8cm, then the red LED will be turned on, otherwise it is turned off. This threshold has been found empirically by filling the can while the sensor was working.

- In the ”irSensorAndStepper” function the rotation speed of the motor is set at 15 RPM, enough to open and close the can’s lid. Since the output of the IR sensor is analog, the ”analogRead” function is used to convert the sensor’s output into a digital value. This value is printed in the ”Serial Monitor”. Between the measuring and checking if the can’s lid must be opened or closed or not moved at all, there’s a delay to not stress the sensor. If the can’s lid is closed and the sensor’s output is bigger than 630 the motor’s shaft will move clockwise in order to open the can’s lid, the ”opened” variable becomes true and a delay of 1.5 seconds in introduced in order to keep the lid opened while the garbage is thrown in the can. If the can’s lid is opened and the sensor’s output is less or equal to 440 the motor’s shaft will move counterclockwise to close the can’s lid and the ”opened” variable becomes false. If none of the above conditions are met, then the can’s lid doesn’t move at all. The motor’s shaft is rotated using the ”step” function which is a blocking one and this means that other pieces of code will not execute while the shaft rotates. The 2 thresholds (440 and 630) related to the sensor have been found empirically by moving objects in front of the sensor while it was working. There are 2 thresholds, because of the noise that makes the sensor return higher values in some moments despite that there isn’t anything moving in front of it. The number of steps the motor takes have also been found empirically by trying different values until the values that lead to the full opening and closing of the lid have been found.
- In the ”loop” function the ”ultrasonicSensorMeasurement” and ”irSensorAndStepper” functions are called. The ”ultrasonicSensorMeasurement” function is first called in order to check if the can is full when the board is powered up.

4. Problems and solutions

I encountered the following 2 problems:

I. I initially used an Intel Galileo Gen 2 board, but it was broken that the microUSB port used to connect the board to a PC doesn’t work. As a replacement I used an Arduino Uno R3 board, because it has the same layout and similar specifications.

II. I initially used a Plusivo 23HS5628 stepper motor, but I changed it with a 5V 28BYJ-48 stepper motor with an ULN2003 driver, because I found that 5V were not enough to supply the first one and the second one is powerful enough to open the lid.

5. Conclusions

Despite the problems that I have encountered I managed to make the intelligent trash can work as intended. Following this project:

- I became familiar with the Arduino Uno R3 board and Arduino IDE.
- I learnt how to program an Arduino board using the afferent language.
- I learnt how to use sensors and stepper motors for an embedded software project.

6. Source code

```
#include <Stepper.h>


#define LED 2

#define TRIG 3

#define ECHO 4


const uint16_t stepsPerRevolution = 2038;

Stepper stepperMotor = Stepper(stepsPerRevolution, 8, 10, 9, 11);  // IN1, IN3,
IN2, IN4


void setup() {

    Serial.begin(9600);

    pinMode(LED, OUTPUT);

    pinMode(TRIG, OUTPUT);

    pinMode(ECHO, INPUT);

}




void ultrasonicSensorMeasurement() {

    digitalWrite(TRIG, LOW);

    delay(150);

    digitalWrite(TRIG, HIGH);

    delay(300);

    digitalWrite(TRIG, LOW);

    double t = pulseIn(ECHO, HIGH);

    double distance = t * 0.017015;

    Serial.print("distance: ");

    Serial.println(distance);

    if (distance < 7.8)   // there's garbage in the way, therefore the can is full

        digitalWrite(LED, HIGH);

    else
```

```
        digitalWrite(LED, LOW);
}

bool opened = false;

void irSensorAndStepper() {
    stepperMotor.setSpeed(15);
    uint16_t irOutput = analogRead(A0);
    Serial.print("irOutput: ");
    Serial.println(irOutput);
    delay(100);
    if (opened == false && irOutput > 630) {   // the can's lid opens
        stepperMotor.step(7000);
        opened = true;
        delay(1500);
    }
    else if (opened == true && irOutput <= 440) {   // the can's lid closes
        stepperMotor.step(-7000);
        opened = false;
    }
    else {}   // the can's lid doesn't move
}


void loop() {
    ultrasonicSensorMeasurement();
    irSensorAndStepper();
}
```